Praktikum Sprac himplementierung mit Werkzeugen

Uwe Kastens WS 1999/2000

Ziele:

Die Teilnehmer sollen lernen

• anwendungsspezifisc he Sprac hen zu entwerfen,

• Implementierungsaufgaben dafür zu erkennen, einzuordnen

und zu beschreiben,

• Spezifikationstec hniken für die zentralen Werkzeuge von

Eli anzuwenden,

• das Eli-System wirksam einzusetz en und zu bedienen.

Sie sollen einselbstdefinier tes Anwendungspr ojekt durchführen können.

Eli-1.0

Beispiele für Pr ojektthemen

• Generator für Datenstrukturen
• Literatur-Report-Generator
• Steuer sprac he für Meßrec hner
• Makrosprac he zum Test von Instrumenten
• Programmstrukturen nac h HTML übersetzt
• Sprac herweiterung en (z. B. Occam)
• Notensc hrift-Generator
• Interpretierer für Entsc heidungstabellen
• Tischrechner
• Lindenmeyer-Grammatiken zur Bilderz eugung

Eli-1.1

Sprac hen in der Software-Entwic klung

Spezifikations- und Modellierungssprac hen:

SDL Specification and Description Langua ge (CCITT):

UML Unified Modeling Language:

Eli-1.1a

Sprac hen in der Software-Entwic klung

Domain-Specific Langua ge (DSL):

Sprache zugeschnitten auf ein spezielles Anwendungsgebiet

Application Generator s:

Implementierung einer DSL durch einen Programmgenerator

Beispiele:

• Simulation mec hatronisc her Reg elsysteme
• Roboter steuerung
• Meßdatenerfassung
• Steuerung v on Prüfständen für Kfz-Instrumente
• Literatur-Report-Generator:

string name =   InString "Which author?";
int since =     InInt "Since which year?";
int cnt = 0;

"\nPapers of ", name, " since ", since, ":\n";
[ SELECT name <= Author && since <= Year;
  cnt = cnt + 1;
  Year, "\t", Title, "\n";
]
"\n", name, " published ", cnt, "papers.\n";

aus U. Kastens: Construction of Application Generators Using Eli, Workshop on Compiler Techniques for Application Domain Languages ..., Linköping, April 1996

zum Thema:

C.W. Krueger: Software Reuse , ACM Computing Surveys 24, June 1992

Conference on DSL (USENIX), Santa Babara, Oct. 1997

ACM SIGPLAN Workshop on DSL (POPL), Paris, Jan 1997

US-3

Sprac hen in der Software-Entwic klung

Programmier sprac hen als Quellsprac hen:

• Programmanal yse

Aufrufgraphen, Ablaufgraphen, Variablenabhängigkeiten z. B. für Year-2000-Problem

• Strukturerkenn ung

z. B. für Reengineering

Programmier sprac hen als Zielsprac hen generiert aus

• Spezifikationen (SDL, OMT, UML)
• graphisc hen Strukturmodellen
• DSL, Application Generator

=> Übersetz eraufgabe: Sour ce-to-Sour ce-Über setzung

US-4

Praktikumsplan

Termin Vorlesung Tutorium Vor- u. Nacharbeit

1 Eli: Aufgabenzerlegung Novice-Guide

2 Novice-Guide durcharbeiten

3 konkrete u. abstrakte Syntax

Berechnungen im Baum Computations in Trees

4 Programmanalyse I

5 Modulbenutzung,

Namensanalyse,

Eigenschaften von Objekten

6 Programmanalyse II

Core-Spezifikation

7 Minisprache Core vorstellen Minisprache Core erweitern

8 Zieltext erzeugen Core nach C übersetzen

9 Projekt: Struktur-Generator vorstellen Projekt: Struktur-Generator bearbeiten

10 - 14 Eigene Projekte definieren und bearbeiten

15 Vorstellung der Projektergebnisse

Eli-1.2

Compilation Tasks f or Sour ce-to-Sour ce Translation

text in a domain specific langua ge

application pr ogram (HLL)

Structuring

Translation

Encoding

Lexical analysis

Syntactic analysis

Semantic analysis

Transformation

Code generation

Assembly

Scanning Conversion

Parsing Tree construction

Name analysis

Type analysis

Data mapping Action mapping

Execution-order Register allocation Instructionselection

Internal addressing External addressing Instructionencoding

Waite, Carter: An Introduction to Compiler Construction, Harper Collins, 1992

Eli-1.3

Eli provides solution methods

generating tools

reusab le precoined solutions f or common tasks

extensible by domain specific routines

Lexical analysis

Syntactic analysis

Semantic analysis

Transformation scanner generator

parser generator

tree construction generator

attribute evaluator definition table generator

pattern based text generator input processing

token sets

encoding &conversion

name analysis modules type analysis modules

generator

Eli-1.4

Integrated tool set Eli

Eli knows how to

•drive the g enerator s
•obtain the precoined solutions
•compose the translator
•relate the user specifications

User specifications f or sour ce-to-sour ce processor s:

conte xt-free grammar

specify object properties

specify target text patterns select non-literal tokens

use name analysis module use type analysis module associate roles to tree grammar

compute in tree conte xts

compose target structure token descriptions

Eli-1.5

Examples for Specification Fra gments

• conte xt-free grammar
   OutputItems:    OutputItem // ','.

• select non-literal token specification
   Ident:          C_IDENTIFIER

• select name anal ysis module
   $/Name/CScope.gnrc:inst

• associate r oles to tree grammar symbols
   SYMBOL DefIdent INHERITS IdDefScope END;
   SYMBOL UseIdent INHERITS IdUseEnv END;

• specify object pr operties
   Kind: DefTableKey;

• compute in tree conte xt
   RULE: Statement ::= UseIdent '=' Expr ';'
   COMPUTE
   IF (NE (GetKind (UseIdent.Key, VariableKind),
           VariableKind),
     message (ERROR, "variable required",
              0, COORDREF));
   END;

• specify tar get text patterns
   InpCall: "(" $1 string "(" $2 string "))"

• compose tar get structure
   RULE: Expr ::= 'InString' StringLit COMPUTE
     Expr.CCode =
       PTGInpCall ("InString",
                   StringTable(StringLit));
   END;

Eli-1.6

FunnelWeb: Spezifikationsfra gmente zusammenfassen und dokumentieren

@=~
Umschaltzeichen von @ auf Tilde umgestellt
~p maximum_input_line_length = infinity
beliebig lange Zeilen
Die Ausgabedatei ArrayType.con erzeugen:
~O~<ArrayType.con~>~{
Type: 'array' Index 'of' Type.
~}
Noch eine Ausgabedatei:
~O~<ArrayAccess.con~>~{
Variable: Variable '[' Expr ']'.
~}
Eli fügt alle .con Dateien zu einer zusammen.
Dies kann man auch mit fw-Makros tun:
Produktionen für Array-Typen:
~$~<Array type~>==~{
Type: 'array' Index 'of' Type.
~}
Produktionen für Array-Zugriffe:
~$~<Array access~>==~{
Variable: Variable '[' Expr ']'.
~}
Die zusammengesetzte Ausgabedatei:
~O~<Array.con~>~{
~<Array type~>
~<Array access~>
~}
Der Eli-Ausdruck X.fw : fwGen liefert eine
Directory mit den Ausgabedateien.

Eli-1.7

Konkrete Syntax

Eine kontext-freie Grammatik definiert die Struktur der Programme der Sprache. Hieraus wird der Parser generiert. Sie muß deshalb eindeutig sein (und die LALR(1)-Bedingung erfüllen).

Notation an Beispielen:

Prog    ::= Block .
Block   ::= '{' Decl* Stmt+ '}' .
Stmt    ::= Block / Assign ';' / Expr ';' .
Decl    ::= 'int' VarDecls ';' .
VarDecls::= VarDecl // ',' .
VarDecl ::= Ident .

Benannte Nichtterminale: Prog Block Decl Benannte Terminale: Ident Literale Terminale: '{' '}' ';'

EBNF-Konstrukte:

Folge: Decl* nicht-leere Folge: Stmt+ nicht-leere Folge mit Trenner: VarDecl // ',' Alternativen: Block / Assign ';' optional: [ Parameter ] Klammerung: ( ':=' / ':-' )

Hinweise zum Entwurf:

• Syntax einer vorgegebenen Sprache nicht unnötig ändern.
• Syntaktische Struktur soll der semantischen entsprechen.
• EBNF nur, wo keine Berechnungen im Baum erwartet werden.
• Namensanalyse erfordert syntaktische Unterscheidung von

Bezeichnern: definierendes und angewandtes Auftreten.

Eli-2.1

Abstrakte Syntax

Eine kontext-freie Grammatik definiert Strukturbäume als Programmrepräsentation (Baum-Grammatik). Der Strukturbaum wird vom Parser aufgebaut.

Analyse und Transformation werden als Berechnungen in bestimmten Strukturbaum-K ontexten spezifiziert.

Konkrete Pr oduktion:

Decl    ::= 'int' VarDecls ';' .

abstrakte Pr oduktion, Baum-Kontext:

RULE: Decl ::= 'int' VarDecls ';' COMPUTE

Berechnungen für diesen Kontext

END;

Die Knoten des Strukturbaumes repräsentieren Nichtterminale der Baum-Grammatik. Terminale stehen nicht im Baum. Literale Terminale dienen nur zur Identifikation der Produktion.

EBNF-Konstrukte gibt es in der Baum-Grammatik nur sehr eingeschränkt (LISTOF).

Es brauchen nur die Baum-Kontexte angegeben zu werden, zu denen es tatsächlich Berechnungen gibt. Eli-Werkzeuge vervollständigen die Baum-Grammatik automatisch.

Ebenso ergänzen sie Produktionen zur konkreten Syntax, die nur in der Baum-Grammatik stehen.

Der Aufbau des Strukturbaumes durch den Parser wird von Eli automatisch generiert.

Eli-2.2

Überflüssig e Kettenpr oduktionen

Konkrete Produktionen mit gleicher Struktur und Bedeutung zu einem Baum-Kontext zusammenfassen. Kettenproduktionen zwischen äquivalenten Symbolen entfallen.

Konkrete Ausdruc ksgrammatik:

Expr    ::= Expr AddOpr Fact / Fact .
Fact    ::= Fact MulOpr Operand / Operand .
Operand ::= '(' Expr ')' .
Operand ::= Number .
AddOpr  ::= '+' / '-' .
MulOpr  ::= '*' / '/' .

Äquiv alente Symbole:

Expr ::= Fact Operand .
Opr  ::= AddOpr MulOpr .

Baum-Grammatik dazu:

RULE: Expr ::= Expr Opr Expr  END;
RULE: Expr ::= Number         END;
RULE: Opr  ::= '+'            END;
RULE: Opr  ::= '*'            END;
...

Strukturbaum dazu:

Ein Symbolknoten gehört zu zwei benachbarten Kontexten: oberer und unterer Kontext.

Expr

Expr Expr

Expr

Expr

Number Number

Number + *

Opr Opr

Eli-2.3

Abhängige Berechnungen

Der Wert eines Ausdruc ks wird berec hnet und ausg egeben:

ATTR value: int;

RULE: Root ::=  Expr  COMPUTE
   printf ("value is %d\n", Expr.value);
END;

TERM Number: int;

RULE: Expr ::= Number COMPUTE
  Expr.value = Number;
END;

RULE: Expr ::= Expr Opr Expr COMPUTE
  Expr[1].value = Opr.value;
  Opr.left  = Expr[2].value;
  Opr.right = Expr[3].value;
END;

SYMBOL Opr: left, right: int;

RULE: Opr ::=  '+'  COMPUTE
  Opr.value  = ADD (Opr.left, Opr.right);
END;

RULE: Opr ::=  '*'  COMPUTE
  Opr.value = MUL (Opr.left, Opr.right);
END;

Eli-3.1

Vor- und Nachbedingung en von Berechnungen

Ausgabe v on Ausdrüc ken in Postfix-Form:

RULE: Root ::= Expr COMPUTE
  Expr.print = "yes";
  printf ("\n") <- Expr.printed;
END;
RULE: Expr ::= Number COMPUTE
  Expr.printed =
    printf ("%d ", Number) <- Expr.print;
END;
RULE: Opr  ::= '+' COMPUTE
  Opr.printed = printf ("+ ") <- Opr.print;
END;
RULE: Opr  ::= '*' COMPUTE
  Opr.printed = printf ("* ") <- Opr.print;
END;
RULE: Expr  ::= Expr Opr Expr COMPUTE
  Expr[2].print = Expr[1].print;
  Expr[3].print = Expr[2].printed;
  Opr.print = Expr[3].printed;
  Expr[1].printed = Opr.printed;
END;

Die Attribute print und printed haben keinen Wert

(Typ VOID).

Sie beschreiben Zustände als Vor- und Nachbedingungen für Berechnungen:

Expr.print: Postfix-Ausgabe bis vor diesem Knoten ist

erledigt.

Expr.printed: Postfix-Ausgabe bis einschließlich dieses

Knotens ist erledigt.

Eli-3.2

Abhängigkeitsm uster: CHAIN

CHAIN spezifiziert left-right-depth-fir st Abhängigkeit. Berechnungen werden in die CHAIN „eingehängt“.

Ausgabe von Ausdrücken in Postfix-Form:

CHAIN print: VOID;
RULE: Root ::= Expr COMPUTE
  CHAINSTART HEAD.print = "yes";
  printf ("\n ") <- TAIL.print;
END;
RULE: Expr ::= Number COMPUTE
  Expr.print =
    printf ("%d ", Number) <- Expr.print;
END;
RULE: Opr ::= '+' COMPUTE
  Opr.post = printf ("+") <- Opr.pre;
END;
RULE: Expr ::= Expr Opr Expr COMPUTE
  Opr.pre = Expr[3].print;
  Expr[1].print = Opr.post;
END;

Wertberechnung entlang einer CHAIN, z. B. Variablenadressen:

CHAIN RelAdr: int;
RULE: Block ::= '{' Sequence '}' COMPUTE
  CHAINSTART HEAD.RelAdr = 0;
END;
RULE: Definition ::= 'define' Ident COMPUTE
  Definition.RelAdr =
    ADD (Definition.RelAdr, VariableSize);
END;

Die Kontexte, in denen die CHAIN nicht benutzt wird, brauchen nicht erwähnt zu werden!

Eli-3.3

Abhängigkeitsm uster: INCLUDING

Ein Attribut wird tiefer im Unterbaum ben utzt.

INCLUDING Block.depth bezieht sich auf das depth-Attribut

des nächsten Oberknotens Block.

Berechnung der Sc hachtelungstief e von Blöcken:

ATTR depth: int;
RULE: Root ::= Block COMPUTE
  Block.depth = 0;
END;
RULE: Statement ::= Block COMPUTE
  Block.depth = ADD (INCLUDING Block.depth, 1);
END;
TERM Ident: int;
RULE: Definition ::= ‘define' Ident  COMPUTE
  printf ("%s defined on depth %d\n ",
           StringTable (Ident),
           INCLUDING Block.depth);
END;

Die Kontexte zwischen der Definition und der Benutzung der

INCLUDING-Attribute brauchen nicht erwähnt zu werden!

Eli-3.4

Abhängigkeitsm uster: CONSTITUENTSfür Zustände

Zusammenfassen v on Attributen aus Unterbaum

CONSTITUENTS Definition.DefDone bezieht sich auf die
DefDone-Attribute allerDefinition-Knoten im Unterbaum.

Ausgabe aller Definitionen v or allen Anwendung en:

RULE: Block ::= '{' Sequence '}' COMPUTE
  Block.DefDone =
    CONSTITUENTS Definition.DefDone;
END;
RULE: Definition ::= 'Define' Ident COMPUTE
  Definition.DefDone =
    printf ("%s defined in line %d\n",
            StringTable(Ident), LINE);
END;
RULE: Usage ::= 'use' Ident COMPUTE
   printf ("%s used in line %d\n ",
           StringTable(Ident), LINE),
   <- INCLUDING BLOCK.DefDone;
END;

Die Kontexte zwischen der Definition und der Benutzung der

CONSTITUENTS-Attribute brauchen nicht erwähnt zu werden!

Eli-3.5

Abhängigkeitsm uster: CONSTITUENTSfür Werte

Zusammenfassung von Werten erfordert

   CONSTITUENTS ... WITH (Type, BinFct, UnFct, ZeroFct)

Type Typ des Ergebnis und der propagierten Werte
BinFct verknüpft zwei Werte; sollte assoziativ sein
UnFct erzeugt einen Wert aus dem Attributwert
ZeroFct erzeugt einen "neutralen" Wert

Ausgabe der Anzahl der Anwendung en:

ATTR Count: int;
RULE: Block ::= '{' Sequence '}' COMPUTE
  printf  ("%d uses occurred\n",
           CONSTITUENTS Usage.Count
           WITH (int, ADD, IDENTICAL, ZERO));
END;
RULE: Usage ::= 'use' Ident COMPUTE
  Usage.Count = 1;
END;

Die Werte werden gemäß der Baumstruktur zusammengefaßt.

Eli-3.6

Berechnungen zu Symbolen

Berechnungen können Symbolen zug eordnet wer den, wenn sie für jeden R ULE-Kontext gleich sind, z. B.

SYMBOL Expr COMPUTE
  printf("expression value %d in line %d\n", THIS.value, LINE);
END;

Symbolberechnungen können auch INCLUDING-,

CONSTITUENTS-, oder CHAIN-Konstrukte enthalten:

SYMBOL Block COMPUTE
   printf ("%d uses occurred\n",
           CONSTITUENTS Usage.Count SHIELD Block
           WITH (int, ADD, IDENTICAL, ZERO));
END;
SYMBOL Usage COMPUTE
  SYNT.Count = 1;
END;

SYNT.a bzw. INH.a gibt an, daß die Berechnung zum unteren

bzw. oberen Kontext des Symbols gehört.

Berechnungen imRULE-Kontext überschreiben Berechnungen im SYMBOL-Kontext für dasselbe Attribut.

Verwendung für Rekursionsanfänge oder für Ausnahmen von der Symbolberechnung:

SYMBOL Block COMPUTE
  INH.depth = ADD (INCLUDING Block.depth, 1);
END;
RULE: Root ::= Block COMPUTE
  Block.depth = 0;
END;

Eli-4.1

Berechnungsmodule mit CLASS-Symbolen

Wiederverwendung v on Berec hnungsm ustern

Berechnungen werden neuen CLASS-Symbolen zugeordnet und durch INHERITS an Baum-Symbole gebunden.

ATTR Sym: int;
CLASS SYMBOL IdOcc COMPUTE SYNT.Sym = TERM; END;
SYMBOL DefIdent INHERITS IdOcc END;
SYMBOL UseIdent INHERITS IdOcc END;

Zusammenwirkende CLASS-Symbole: z. B. Auftreten eines syntaktischen Konstruktes zählen:

CHAIN Occurrence: int;
ATTR OccNo, TotalOccs: int;
CLASS SYMBOL OccRoot COMPUTE
  CHAINSTART HEAD.Occurrence = 0;
  THIS.TotalOccs = TAIL.Occurrence;
END;
CLASS SYMBOL OccElem COMPUTE
  SYNT.OccNo = THIS.Occurrence;
  THIS.Occurrence = ADD (SYNT.OccNo, 1);
END;

Benutzung:

SYMBOL Block INHERITS OccRoot END;
SYMBOL Definition INHERITS OccElem END;
SYMBOL Statement INHERITS OccRoot END;
SYMBOL Usage INHERITS OccElem END;

Auf diesem Prinzip beruht die Benutzung Bibliotheksmodulen, z. B. zur Namensanalyse.

Eli-4.2

Bibliotheksmodule allg emein

Bibliotheksmodule mac hen vorgefertigte Lösung en für häufig auftretende A ufgaben verfügbar .

1. Modul in der Dokumentation identifizieren, Beschreibung lesen

2. Modul instantiieren:

$/Prop/Unique.gnrc : inst

3. Symbolrollen Grammatiksymbolen zuordnen und Attribute benutzen:

 SYMBOL Program  INHERITS RangeUnique END;
 SYMBOL DefIdent INHERITS Unique COMPUTE
   IF (NOT (THIS.Unique),
   message (ERROR,
            CatStrInd
             ("identifier is multiply defined: ",
              THIS.Sym),
            0, COORDREF));
 END;

4. Wenn nötig Modul-Implementierung beschaffen und lesen:

-> $elipkg/Prop/Unique.gnrc : inst : viewlist

Eli-4.3

ADT Listen

Generisc her Modul zur Listenimplementierung

  $/Adt/List.gnrc+instance=DefTableKey +referto=deftbl :inst

Hier: ElementtypDefTableKey definiert in Dateideftbl.h

Es werden der Typ DefTableKeyList und Funktionen wie

ConsDefTableKeyList, HeadDefTableKeyList,
TailDefTableKeyList, ...und viele andere

als C-Modul verfügbar.

Lido-Unterstützung zur Listenben utzung:

$/Adt/LidoList.gnrc+instance=DefTableKey+referto=deftbl :inst

Listen in Baumkontexten aufbauen:

   RULE: FunctionHead ::= '(' Parameters ')' TypeDenoter
   COMPUTE
      ...ResetParamTypes (FunctionHead.Type,
                          Parameters.DefTableKeyList) ...
   END;
   SYMBOL Parameters INHERITS DefTableKeyListRoot END;
   SYMBOL Parameter  INHERITS DefTableKeyListElem END;
   RULE: Parameter ::= TypeDenoter DefIdent COMPUTE
     Parameter.DefTableKeyElem = TypeDenoter.Type;
   END;

Listen in Baumkontexten abbauen entsprechend.

Weitere ADT-Module: BitSet, Stack

Eli-4.4

Namensanal yse (Bezeichneridentifikation)

Bezeichner benennen (statische) Programmobjekte.

Definition für Bezeichner b führt neues Programmobjekt ein und bindet es an den Bezeichner.

Die Bindung gilt in einem bestimmten Bereich des Programms: Gültigkeitsbereic h der Definition.

Aufgabe: jedem Auftreten eines Bezeichners den Schlüssel des Programmobjektes zuordnen. (konsistente Umbenennung)

Grundlage: Gültigkeitsregeln der Sprache (scope rules)

Verdeckungsregeln für Sprachen mit geschachtelten Strukturen:

• Algol-Regel: Die Definition eines Bezeichners b gilt imganzen

kleinsten sie umfassenden Abschnitt, aber nicht in darin enthaltenen Abschnitten mit einer Definition von b.

• C-Regel: Die Definition eines Bezeichners b gilt im kleinsten

sie umfassenden Abschnitt von der Definitionsstelle an , aber nicht in darin enthaltenen Abschnitten mit einer Definition von b von der Definitionsstelle an.

Abschnitt je nach Sprache: Block. Prozedur, Modul, ...

Implementierung : Operationen des Umgebungsmoduls in Baumkontexten aufrufen.

Eli-5.1

Umgebungsmodul

implementiert den abstrakten DatentypEnvironment :

hierarchisch geschachtelte Mengen von Bindungen

(Bezeichner, Schlüssel)

Funktionen:

NewEnv () erzeugt ein neues Environment e als Wurzel.

NewScope (e1) erzeugt neues Environment e2. Alle Bindungen aus e1 sind auch in e2, falls sie nicht durch Bindungen in e2 verdeckt werden.

DefineIdn (e, b) fügt Bindung (b, k) in e ein, falls sie dort noch nicht exisitiert; k ist dann ein neuer Objektschlüssel. Liefert in jedem Fall den an bgebundenen Schlüssel k.

KeyInEnv (e, b) liefert den Schlüssel k eines Paares (b, k) aus e (einschließlich aller e_i, aus denen e erzeugt wurde). Liefert NoKey, falls kein solches Paar existiert.

KeyInScope (e , b)liefert den Schlüssel eines Paares (b, k), falls unmittelbar in e enthalten, NoKey sonst.

Eli-5.2

Umgebungsoperationen im Baum

Operationen in Baumkontexten und Reihenfolge ihrer Ausführung modelliert Gültigkeitsregeln.

Wurzelkontext Root:

Root.Env = NewEnv ();

Abschnitte Range, die Definitionen enthalten können:

Range.Env =
  NewScope(INCLUDING (Range.Env, Root.Env));
           nächst umgebende Range oder Root

definierendes A uftreten eines Bez eichners IdDefScope:

IdDefScope.Key =
  DefineIdn(INCLUDING Range.Env,
            IdDefScope.Symb);

angewandtes A uftreten eines Bez eichners IdUseEnv:

IdUseEnv.Key =
  KeyInEnv (INCLUDING Range.Env,
            IdUseEnv.Symb);

Vorbedingung en je nac h Gültigkeitsreg eln: Algol-Regel: Vorbedingung fürKeyInEnv: alle DefineIdn aller umgebenden Ranges sind ausgeführt.

C-Regel: KeyInEnv und DefineIdn werden in der Reihenfolge des Programmtextes ausgeführt (links-abwärts).

weitere Prüfung en an dem Er gebnis der Bez.identifikation:

• keine Anwendung ohne Definition
• keine 2 Definitionen für einen Bezeichner in einem Range
• keine Anwendung vor der Definition (Pascal)
• Vorwärtsdefinitionen
• ...

Eli-5.3

Module zur Namensanal yse

Symbolrollen:

Wurzel der Grammatik:

  SYMBOL Program INHERITS RootScope END;

Abschnitte mit Definitionen:

  SYMBOL Block INHERITS RangeScope END;

Definierter Bezeichner:

  SYMBOL DefIdent INHERITS IdDefScope END;

Angewandte Bezeichner:

   SYMBOL UseIdent INHERITS IdUseEnv END;
   SYMBOL TypeUseIdent INHERITS IdUseEnv END;

Benutzbare Attrib ute:

   DefIdent.Key, UseIdent.Key
   Program.Env, Block.Env

Instantiierung für Algol-Scope-Rules:

   $/Name/AlgScope.gnrc : inst

Instantiierung für C-Scope-Rules:

   $/Name/CScope.gnrc : inst

Instantiierung für neuen Namensraum

   $/Name/AlgScope.gnrc +instance=Label :inst

Symbolrollen:LabelRootScope, LabelRangeScope, ...

Instantiierung für zusätzlic hes Key-Attribut:

   $/Name/AlgScope.gnrc +instance=Label +referto=Lab :inst

Attributnamen: DefIdent.LabKey, UseIdent.LabKey

Eli-5.4

PDL: Generator für Definitionsmodule

zentrale Datenstruktur ordnetProgrammobjekten z. B. Typen, Variablen, ... Eigensc haften zu z. B. Typ einer Variablen, Elementtyp eines Array-Typs.

Objekte werden durch Schlüssel (key) identifiziert.

Operationen:

NewKey ( ) liefert neuen Schlüssel

ResetP (k, v) setzt zum Schlüssel k die

Eigenschaft P mit Wert v

SetP (k, v, d) setzt (ersetzt) zum Schlüssel k die

Eigenschaft P mit Wert v (bzw. d)

GetP (k, d) liefert Wert der EigenschaftP

des Schlüssels k; liefert d, falls P zu k nicht gesetzt ist

Aufrufe: abhängige Operationen im Baum

Implementierung: z. B. Liste von Eigenschaften zu jedem Schlüssel

Generierung des Definitionsmoduls : Aus Spezifikationen

Eigenschaftsname: Eigenschaftstyp;

werden FunktionenReset, Set, Get generiert.

Eli-5.5

Richtlinien zum Entwurf v on

Berechnungen im Baum

1.Zerlege die Aufgabe inTeilaufgaben , die klein genug sind, um durch wenige der u. g. Spezifikationsmuster gelöst zu werden. Entwickle für jede Teilaufgabe ein.lido-Fragment und erläutere es im umgebenden .fw-Text.

2.Arbeite denzentralen Aspekt der T eilaufgabe heraus und bilde ihn auf einen der folgenden Fälle ab: A. Der Aspekt wird in natürlicher Weise durch Eigensc haften von einig en Programmkonstrukten beschrieben, z. B. Typen vo Ausdrücken, Schachtelungstiefe von Blökken, Übersetzung der Anweisungen eines Blockes.

B. Der Aspekt wird in natürlicher Weise durch Eigensc haften von Programmobjekten beschrieben, z. B. Relativadresse von Variablen, Benutzung von Variablen vor der Definition.

Entwirf die Berechnung wie für A oder B beschrieben.

3.Im Schritt 2 können Anforderungen an weitere Aspekte der Teilaufgabe entstehen (benutzte Attribute, deren Berechnung noch nicht entworfen ist): Wiederhole Schritt 2 für diese.

Eli-5.6

A: Eigensc haften von Programmkonstrukten

Bestimme den Typ der Werte, die die Eigenschaft beschreiben. Führe Attribute diese T yps zu allen Symbolen ein, die die Programmkonstrukte repräsentieren. Prüfe, welcher der folgenden Fälle für die Bestimmung der Eigenschaft am besten zutrifft.

A1:Jeder untere Kontext bestimmt die Eigenschaft in unterschiedlicher Weise: EntwirfRULE-Berechnungen.

A2:Wie A1; aberoberer Kontext.

A3:Die Eigenschaft kannunabhängig v on den R ULE-Kon-texten berechnet werden, wobei nur Attribute des Symbols oder über INCLUDING, CONSTITUENT(S), CHAIN erreichbare Attribute verwendet werden: Entwirf eine untere (SYNT) SYMBOL-Berechnung.

A4:Wie A3. Aber es gibtwenige Ausnahmen , wo entweder untere oder obere (nicht beide!) RULE-Kontexte die Eigenschaft abweichend bestimmen: Entwirf eine obere (INH) oder untere (SYNT)SYMBOL-Berechnung und überschreibe sie in solc hen RULE-Kontexten.

A5:Wie A4; aber fürrekursive Symbole : Die Fundierung der Rekursion ist die Ausnahme aus A4, z. B. Schachtelungstiefe von Blöcken.

Wenn keiner der Fälle paßt, muß die Modellierung der Eigenschaft überdacht werden. Sie könnte zu komplex sein und eine weitere Zerlegung erfordern.

Eli-5.7

B: Eigensc haften von Programmobjekten

Vorbereitung: Programmobjekte werden meist durch Bezeichner im Programm benannt. In diesem Fall verwendet man einen Modul zurNamensanal yse, um Programmobjekte (repräsentiert durchDefTableKeys) and die Auftreten von Bezeichnern zu binden. Damit hat jeder Symbolknoten der Bezeichner ein Key-Attribut, dessen Wert das Programmobjekt identifiziert.

Berechnungen, die die Eigenschaft des Programmobjektes setz en oder lesen , sollten vorzugsweise im unteren oder oberen Kontext des Symbols für den Bezeichner plaziert werden, statt die Key-Werte im Baum weiterzureichen. So wird die Zugehörigkeit der Berechnung zum Programmobjekt deutlicher. Wenn möglich solltenSYMBOL-Berechnungen verwendet werden (siehe A).

Im allgemeinen benötigt man folgende Entwurfsschritte:

1.BestimmeNamen und Typ der Eigensc haft und spezifiziere sie für PDL.

2.Bestimme dieKontexte und die Operation für das Setz en der Eigenschaft.

3.Bestimme dieKontexte für das Lesen der Eigenschaft.

4.Bestimme dieAbhängigkeiten zwisc hen (2) und (3). In einfachen Fällen ist dies: "alle Setzen vor jedem Lesen".

5.Spezifiziere (2), (3) und das Abhängigkeitsmuster (4).

Eli-5.8

Minisprache Core

Kleine imperative Programmiersprache, statisch typisiert. Spezifikation der Analyse liegt vollständig vor.

Ziele:

• Spezifikation der T ypanalyse kennenlernen

-> Spezifikation verstehen, Sprache erweitern

• Technik der Sour ce-To-Source-Über setzung erlernen

-> Abschnitt 7, Spezifikation vervollständigen

Core Beispielpr ogramm:

program
let
     a:    int;
     x, y: real;
     i = 10;
in
     input a;
     x := 3.5;
     y := a * i + x;
     output y;
end

Sprac hkonstrukte:

• Block
• Variablen- und Konstantendeklarationen
• Zuweisungen
• E/A Anweisungen
• Ausdrücke

Erweiterung en

• Ablaufstrukturen
• geschachtelte Blöcke
• weitere Operatoren

Eli-6.1

Core: Struktur , Namensanal yse

Konkrete Syntax:

Program:        Source.
Source:         'program' Block.
Block:          'let' Declarations 'in'
                      Statements 'end'.
Declarations:   Declaration*.
Statements:     Statement*.

Declaration:    DefIdents ':' TypeDenoter ';'.
DefIdents:      DefIdent // ','.
DefIdent:       Ident.
TypeDenoter:    UseTypeIdent.
UseTypeIdent:   Ident.

Statement:      UseIdent ':=' Expression ';'.
Statement:      'input' UseIdent ';'.
Statement:      'output' Expression ';'.
...
Operand:        UseIdent.
Operand:        IntLiteral.
Operand:        FloatLiteral.
UseIdent:       Ident.

Ausdruckssyntax folgt später. Symbolnotation wie in Pascal.

Namensanal yse (siehe Folie 5.4):

• prüfen: alle Bezeichner definiert, alle Definitionen eindeutig
• Bezeichner int, real mit ModulPreDefine vordefinieren:
   PreDefKey ("int", intKey)
   PreDefKey ("real", realKey)

Eli-6.2

Aufgaben und Begriffe der Typanalyse

Typisierte Programmobjekte (z. B. Variable)

• haben Typ als Eigensc haft;
• Typ wird durch Definition des Objektes bestimmt,
• Typ wird bei Benutzungen verwendet (Prüfung, Analyse)

Typisierte Programmkonstrukte (z. B. Ausdrücke)

• haben Typ als Eigensc haft (Attribut);
• sprachspezifische Regeln zur Bestimmung und Prüfung der

Typen

Typen sind Pr ogrammobjekte

• Sie haben charakterisierende Eigenschaften; diese können

wieder Typen enthalten.

• vordefinier te Typen (z. B. int, real)
• Typen, die durchTypangaben im Programm eingeführt werden

(z. B. record ... end)

• Typbezeichner benennen einen Typ;

Bindung Name - Typ durchTypdefinition; mehrere Namen können an denselben Typ gebunden sein

Relationen über T ypen

• zwei Typen sind gleich,

wenn sie dasselbe Typobjekt sind oder benennen, oder wenn sie sprachspezifische Regeln erfüllen

• ein Typ tn ist verträglich mit einem Typ tw,

wenn Werte von tn als Werte von tw aufgefasst oder in solche umgewandelt werden können (Coercion).

• evtl. weitere sprachspezifische Relationen über Typen

Eli-6.3

Typrepräsentation des Moduls Typing

Typen werden durch DefTableKeys repräsentiert

NoKey: fehlerhafter, unbekannter Typ.

Relation Defer bindet Typ-Keys an die Keys von Typnamen:

Indirekte Bindung en werden erst bei Anwendungen des Typs aufglöst; vermeidet Analysezyklen:

Die FunktionTransDefer angewandt auf einen Typnamens-Key oder Typ-Key liefert den Typ-Key am Ende der Defer-Kette. Voraussetzung: alle Defer-Relationen existieren.

Defer-Ketten werden vom Modulnicht zyklisch erzeugt.

Eigensc haften von Typen werden immer den Typ-Keys (am Ende der Defer-Kette) zugeordnet.

Typ-Keys werden durch die EigenschaftIsType von anderen Keys unterschieden.

int intKey intType

Defer

type  t   =   record ... end

Defer

type  t    =   s;   s   =   record ... end

Defer Defer

Eli-6.4

Typisierte Objekte

Programmobjekte (Variable, Parameter, usw.) haben einen Typ als Eigenschaft TypeOf.

Symbolrollen des Moduls Typing

• setzen die TypeOf-Eigenschaft (TypedDefId)
• im Kontext von Definitionen (TypedDefinition) und
• machen sie bei Anwendungen (TypedUseId) als AttributType

verfügbar (Wert: Typ-Key am Ende der Defer-Kette)

• kennzeichnet Wurzel der Grammatik (RootType)

Zusammenwirken der Rollen garantiert die Reihenfolge für das Setzen und Lesen von TypeOf und Benutzen der Defer-Kette.

Beliebigeweitere Eigensc haften können zusammen mit dem Typ gesetzt werden. Sie sind zusammen mit dem Typ verfügbar

SYMBOL DefIdent INHERITS TypedDefId COMPUTE
  SYNT.GotType = ResetKind (THIS.Key, ...);
END;
SYMBOL UseIdent INHERITS TypedUseId COMPUTE
  SYNT.Kind = GetKind (THIS.Key, ...)
    <- THIS.Type;
END;

          TypedDefId

  TypedDefId

            TypedDefinition

                                    TypedUseId

                        TypAngabe

                .Type =

                           .Type =

                                      .Type

Eli-6.5

Typnotationen

Typangaben z. B. Deklarationen, Signaturen, usw. sind angewandte Typbezeichner (RolleTypeDefUseId) oder Konstrukte , die einen Typ besc hreiben (z. B. Record, Array) (RolleTypeDenotation)

Die RolleTypeDenotation bildet einneues T ypobjekt (Type- Key). BeliebigeEigensc haften des T yps können in diesem Kontext gesetzt werden. Das Attribut GotType garantiert ihre Verfügbarkeit an Anwendungsstellen.

SYMBOL ArrayType INHERITS TypeDenotation END;
RULE: ArrayType ::= TypAngabe '[' Number ']'
COMPUTE
  ArrayType.GotType =
    ORDER (ResetElemType (ArrayType.Type,
                         TypAngabe.Type),
           ResetElemNo (ArrayType.Type, Number));
END;

BeiTypbezeichnern als T ypangabe ( TypeDefUseId) wird der Namens-Key als Typ verwendet; die Defer-Kette wird erst bei Anwendungen aufgelöst.

BeiAnwendung en von Typeigensc haften , die wieder Typen sind, muß TransDefer angewandt werden; Vorbedingung: <- INCLUDING RootType.GotType ist hier nicht nötig, da indirekt durchVariable[2].Type.

RULE: Variable ::= Variable '[' Expr ']' COMPUTE
  Variable[1].Type =
     TransDefer (GetElemType (Variable[2].Type,
                             NoKey));
END;

Eli-6.6

Überladene Operatoren

Ein Operatorsymbol (Quelloperator ) kann verschiedene Bedeutungen (Zieloperator ) haben, sie werden unterschieden durch die Typen der Operanden. Werkzeug OIL und ModulOperator unterstützen die Analyse.

Ausdruc ksgrammatik , z. B.

Expression: Expression AddOpr Factor / Factor.
Factor:     Factor MulOpr Operand / Operand.
Operand:    MonOpr Operand.

Beschreibung der Quelloperatoren (Zeichen, konkr. Symbol, abstr. Symbol, eindeutiger Name)

SrcOpr ('+', AddOpr, BinOpr, AddKey)
SrcOpr ('*', MulOpr, BinOpr, MulKey)
SrcOpr ('-', MonOpr, UnOpr,  NegKey)

Beschreibung der Zieloperatoren (Quellopr., eindeutiger Name, Signatur, Eigenschaftsinit.)

TgtOpr (AddKey, iAddKey, (intType,intType):intType,    TtStr={"+"})
TgtOpr (AddKey, rAddKey, (realType,realType):realType, TgtStr={"||"})
TgtOpr (MulKey, iMulKey, (intType,intType):intType,    TgtStr={"*"})
TgtOpr (MulKey, bAndKey, (realType,realType):realType, TgtStr={"&&"})
TgtOpr (NegKey, iNegKey, (intType):intType,            TgtStr={"-"})
TgtOpr (NegKey, bNotKey, (realType):realType,          TgtStr={"!"})

Implizite Typanpassung für OIL spezifizieren:

  COERCION cFloat (intType):realType;

Attributierung benutzt Modulrollen BinTgtOpr, ChkOpr:

SYMBOL Expression: Type, ReqType: DefTableKey;
SYMBOL BinOpr INHERITS BinTgtOpr, ChkOpr END;
RULE: Expression ::= Expression BinOpr Expression COMPUTE
  BinOpr.LType = Expression[2].Type;
  BinOpr.RType = Expression[3].Type;
  Expression[1].Type = BinOpr.ResType;
  Expression[2].ReqType = BinOpr.LTType;
  Expression[3].ReqType = BinOpr.RTType;
END;

Eli-6.7

Zieltexte erzeugen

Werkzeug PTG: Pattern-based T ext Generator Erzeugung strukturierter Texte in beliebigen Sprachen. Anwendung als Berechnungen im Strukturbaum, und auch in beliebigen C-Programmen.

Prinzip: 1. Ausgabe-P attern mit Einfügestellen spezifizieren:

     ProgramFrame: $
                   "void main () {\n"
                   $
                   "}\n"

     Exit:         "exit (" $ int ");\n"

     IOInclude:    "#include <stdio.h>"

2. PTG generier t Funktionen dazu. Aufrufe liefern Zielstruktur, bottom-up zusammensetzen:

     PTGNode a, b, c;
     a = PTGIOInclude ();
     b = PTGExit (5);
     c = PTGProgramFrame (a, b);

entsprechend mit Attributen im Strukturbaum

3. Ausgabe der Zielstruktur:

     PTGOut (c);
     PTGOutFile ("Output.c", c);

Eli-7.1

PTG-Muster: Wichtige Techniken

Indizierte Einfügestellen Anwendung: gleicher Text wird mehrfac h eingefuegt :

 Module: "module " $1 "begin" $2 "end " $1 ";\n"

Aufruf PTGModule (id, body);

Anwendung: Muster ändern - Aufruf beibehalten : heute Decl: $1 /*type*/ " " $2 /*names*/ ";" morgen Decl: $2 /*names*/ ": " $1 /*type*/ ";" immer Aufruf PTGDecl (tp, ids);

Typisierte Einfügestellen, Muster für Blatt-T exte:

    CarId: $ string "-" $ string "-" $ int

AufrufPTGCarId ("PB", "AB", 127);

String aus Csm-String-Speic her einsetz en:

    CLASS SYMBOL IdOcc COMPUTE
      SYNT.Ptg = PTGId (TERM);
    END;

Optionale P attern-Elemente Anwendung: Listen mit Trenner:

    CommaSeq: $ { ", " } $

AufrufPTGCommaSeq (a, b); setzt Trenner nur ein, wenn weder a noch b gleich PTGNULL ist

Einige nützliche Pattern sind verfügbar in

    $/Output/PtgCommon.fw

Eli-7.2

Zieltext im Strukturbaum zusammensetz en

Benachbarte Kontexte:

ATTR Ptg: PTGNode;
RULE: Statement ::= UseIdent ':=' Expression
COMPUTE
  Statement.Ptg =
    PTGAssign (UseIdent.Ptg, Expression.Ptg);
END;

Zusammenfassen bei ähnlic hen Quell- und Zielstrukturen:

SYMBOL Statements COMPUTE
  SYNT.Ptg =
    CONSTITUENTS Statement.Ptg
    WITH (PTGNode, PTGSeq, IDENTICAL, PTGNull);
END;

Unregelmäßiges Zusammenfassen in link-rec hts Ordnung:

CHAIN ConstPtg: PTGNode;
SYMBOL Block COMPUTE
  CHAINSTART HEAD.ConstPtg = PTGNULL;
END;
RULE: Declaration::= DefIdent '=' IntLiteral ';'
COMPUTE
  Declaration.ConstPtg =
    PTGSeq (Declaration.ConstPtg,
            PTGConstDecl
              (DefIdent.Ptg, IntLiteral.Ptg));
END;

Eli-7.3

Konsistente Umbenenn ung von Bezeichnern

Zu jedem bezeichneten Quellobjekt soll eineindeutig er Zielbezeichner erzeugt werden, der gut lesbar ist.

Die Bindung der Zielbezeichner sollunabhängig v on den Regeln der Zielsprac he sein.

PTG-Muster hängt Nummer an den Quellbezeichner an:

  TransId:  $ string /*src name*/ "_" $ int

PDL-Eigensc haft: Übersetzung des Prog rammobjektes:

  TransId: PTGNode; "ptg_gen.h"

Berechnung zu allen Bezeichnerknoten im Baum:

CLASS SYMBOL IdOcc COMPUTE
  SYNT.Ptg = SetOrGetTransId (THIS.Key, TERM);
END;

C-Modulimplementiert die FunktionSetOrGetTransId:

static int count = 1;
static PTGNode errid = PTGTransId ("_ERR", 0);

PTGNode SetOrGetTransId (DefTableKey k, int sym)
{ PTGNode result = GetTransId (k, PTGNULL);
  if (k == NoKey) return errid;
  if (result == PTGNULL)
  { result =
      PTGTransId (StringTable (sym), count++);
    ResetTransId (k, result);
  }
  return result;
}

Eli-7.4

Projektaufgabe

Generator für Implementierung en von Verbundtypen:

Verbundtypen sollen durch eine einfache Sprache beschr ieben werden, z. B.

import "String.h"
Adresse
     (name:  String;
      plz:   int;
      Stadt: String;
     )

Daraus soll ein ANSI-C-Modul mit Schnittstellen- und Implementierungsdatei erzeugt werden. Er definiert

• den Verbundtyp und seinen P ointertyp,
• eine Erzeugerfunktion newAdresse, die ein Verbundobjekt

erzeugt und den Pointer darauf liefert,

• für jede Komponente ein Zugriffsmakro, z. B.
  #define nameOfAdresse(x) ((x)->name_fld)
• #include-Anweisung en für die benutzten Typen

Vorgehen:

• Zieltext für das Beispiel aufschreiben,
• Konkrete Syntax der Sprache entwerfen,
• Aufgaben der Namensanal yse aufschreiben, spezifizieren,
• notwendige Eigensc haften von Objekten feststellen,

spezifizieren,

• Zieltext-Erzeugung spezifizieren,
• Generator und seine Produkte validieren.

Eli-8.1

Zieltextbeispiel zur Pr ojektaufgabe

Eingabepr ogramm adr:

import "String.h"
Adresse
     (name:  String;
      plz:   int;
      Stadt: String;
     )

Ausgabedatei adr.h:

#include "String.h"

typedef struct __Adresse *Adresse;
typedef struct __Adresse {
  String name_fld;
  int plz_fld;
  String stadt_fld;
} _Adresse;

#define nameOfAdresse(x) ((x)->name_fld)
#define plzOfAdresse(x) ((x)->plz_fld)
#define stadtOfAdresse(x) ((x)->stadt_fld)

extern Adresse newAdresse ();

Ausgabedatei adr.c:

#include "adr.h"
#include <stdlib.h>

Adresse newAdresse ()
{ return (Adresse)malloc(sizeof(_Adresse));
}

Eli-8.2